

9/3.9

Black Box

Dans ce jeu de réflexion, trois trous noirs sont enfermés dans une boîte quadrillée. Vous devez déterminer leur position en dirigeant des faisceaux laser vers la boîte et en observant leur déviation/absorption à partir des règles suivantes :

- toute lumière laser envoyée sur un trou noir est absorbée ;
- toute lumière laser envoyée à proximité d'un trou noir (dans une des cases qui entourent le trou noir) est déviée de 90 degrés.

COMMENT UTILISER LE PROGRAMME

Le programme est proposé dans ses versions Basic et Turbo Pascal. Les deux versions ont un fonctionnement comparable, mis à part que la version Turbo Pascal n'affiche pas d'écran de présentation, ceci afin que les possesseurs de CPC-464 et 664 puissent également en profiter.

Le listing de la version Basic est le suivant :

```

1000 REM =====
1010 REM Black Box
1020 REM =====
1030 REM
1040 GOSUB 1100 'Presentation
1050 GOSUB 1600 'Jeu
1060 GOSUB 4120 'Commentaire de fin de partie
1070 END
1080 REM
1090 REM -----
1100 REM Presentation du jeu
1110 REM -----
1120 REM
1130 DIM t(4,9),r(4,9)
1140 REM
1150 REM -----
1160 REM Remplissage du tableau de jeu
1170 REM -----
1180 REM
1190 FOR i=1 TO 3
1200   a=INT(RND(1)*2)+2
1210   b=INT(RND(1)*6)+2
1220   IF t(a,b)=1 THEN i=i-1 ELSE t(a,b)=1
1230 NEXT i
1240 REM
1250 REM -----
1260 REM Positionnement des indicateurs
1270 REM -----
1280 REM
1290 xp=2:yp=7
1300 REM
1310 REM -----
1320 REM Affichage de la regle du jeu
1330 REM -----
1340 REM
1350 MODE 1
1360 PRINT "           Black Box"
1370 PRINT "           -----"
1380 PRINT
1390 PRINT "Une boite noire renferme trois trous"
1400 PRINT "noirs capables de devier ou d'absorber"
1410 PRINT "le rayon issu d'un tube a effet laser"
1420 PRINT "selon les deux lois suivantes :"
1430 PRINT
1440 PRINT "   ";CHR$(244);"   ";CHR$(244)
1450 PRINT "   | |           |"
1460 PRINT "   | |           |"
1470 PRINT " ";CHR$(247);"-----";CHR$(246);"           ";CHR

```

```

$(245)
1480 PRINT "      ";CHR$(176);"      ----";
1490 PRINT CHR$(246);CHR$(176);CHR$(247);"----"
1500 PRINT " ";CHR$(247);"---- ----";CHR$(246);"
R$(244)
1510 PRINT "      ; ;      ;"
1520 PRINT "      ; ;      ;"
1530 PRINT "      ";CHR$(245);" ";CHR$(245)
1540 PRINT
1550 PRINT"Appuyez sur une touche ..."
1560 ch#=INKEY$
1570 IF ch#="" THEN 1560
1580 RETURN
1590 REM
1600 REM -----
1610 REM Deroulement d'une partie
1620 REM -----
1630 REM
1640 CLS
1650 PRINT "          Black Box"
1660 PRINT "          ----"
1670 FOR i=1 TO 5
1680   FOR j=1 TO 10
1690     LOCATE 8+2*j,6+2*i
1700     PRINT"."
1710   NEXT j
1720 NEXT i
1730 LOCATE 1,20
1740 PRINT"T=Tir, P=Placement, E=Effacement"
1750 LOCATE 11,17
1760 PRINT CHR$(244)
1770 xp=2:yp=7
1780 c=244 'Fleche vers le haut par defaut
1790 REM
1800 REM -----
1810 REM Attente d'une commande
1820 REM -----
1830 REM
1840 ch#=INKEY$
1850 IF ch#="" THEN 1840 'Attente de l'appui sur une touche
1860 ch#=UPPER$(ch#)
1870 IF ch#="T" THEN GOSUB 1930
1880 IF ch#="P" THEN GOSUB 3650
1890 IF ch#="E" THEN GOSUB 3820
1900 IF fin <> 1 THEN 1840
1910 RETURN
1920 REM
1930 REM -----
1940 REM Demande de tir

```

```
1950 REM -----
1960 REM
1970 LOCATE 1,1
1980 PRINT"Servez-vous des touches fleches pour"
1990 PRINT"positionner l'origine du tir"
2000 PRINT"La touche T commande le tir."
2010 ch#=INKEY#
2020 IF ch#="" THEN 2010
2030 IF ASC(ch#)=243 THEN GOSUB 2110 'Vers la droite
2040 IF ASC(ch#)=242 THEN GOSUB 2250 'Vers la gauche
2050 IF ASC(ch#)=241 THEN GOSUB 2390 'Vers le bas

2060 IF ASC(ch#)=240 THEN GOSUB 2530 'Vers le haut
2070 IF UPPER$(ch#)<>"T" THEN 2010
2080 GOSUB 2670 'Essai laser
2090 RETURN
2100 REM
2110 REM -----
2120 REM Deplacement vers la droite
2130 REM -----
2140 REM
2150 IF xp>9 OR yp=1 OR yp=4 OR yp=5 THEN RETURN
2160 LOCATE xp*2+7,yp*2+3
2170 PRINT" "
2180 IF yp=3 THEN yp=yp-1:c=245
2190 IF yp=6 THEN yp=yp+1:c=244
2200 xp=xp+1
2210 LOCATE xp*2+7,yp*2+3
2220 PRINT CHR$(c)
2230 RETURN
2240 REM
2250 REM -----
2260 REM Deplacement vers la gauche
2270 REM -----
2280 REM
2290 IF xp<3 OR yp=1 OR yp=4 OR yp=5 THEN RETURN
2300 LOCATE xp*2+7,yp*2+3
2310 PRINT" "
2320 IF yp=3 THEN yp=yp-1:c=245
2330 IF yp=6 THEN yp=yp+1:c=244
2340 xp=xp-1
2350 LOCATE xp*2+7,yp*2+3
2360 PRINT CHR$(c)
2370 RETURN
2380 REM
2390 REM -----
2400 REM Deplacement vers le bas
2410 REM -----
2420 REM
```

```
2430 IF yp>5 OR (xp<>1 AND xp<>2 AND xp<>10 AND xp<>11) THEN
  RETURN
2440 LOCATE xp*2+7,yp*2+3
2450 PRINT" "
2460 IF xp=10 THEN xp=xp+1:c=247
2470 IF xp=2 THEN xp=xp-1:c=246
2480 yp=yp+1
2490 LOCATE xp*2+7,yp*2+3
2500 PRINT CHR$(c)
2510 RETURN
2520 REM
2530 REM -----
2540 REM Deplacement vers le haut
2550 REM -----
2560 REM
2570 IF yp<4 OR (xp<>1 AND xp<>2 AND xp<>10 AND xp<>11) THEN
  RETURN
2580 LOCATE xp*2+7,yp*2+3
2590 PRINT" "
2600 IF xp=10 THEN xp=xp+1:c=247
2610 IF xp=2 THEN xp=xp-1:c=246
2620 yp=yp-1
2630 LOCATE xp*2+7,yp*2+3
2640 PRINT CHR$(c)
2650 RETURN
2660 REM
2670 REM -----
2680 REM Essai laser
2690 REM -----
2700 REM
2710 REM Effacement des messages sur l'ecran
2720 REM
2730 sc=c 'Sauvegarde de C
2740 LOCATE 1,1
2750 PRINT SPACE$(120)
2760 LOCATE 1,1
2770 PRINT "          Black Box"
2780 PRINT "          -----"
2790 REM
2800 REM -----
2810 REM Calculs de la sortie laser
2820 REM -----
2830 REM
2840 bo=0 'Pas d'absorbtion a priori
2850 de=c-243 'Sens du deplacement
2860 x=yp-2:y=xp-1
2870 ON de GOSUB 3020,3120,3220,2920
2880 IF ff<>1 AND bo<>1 THEN 2870
2890 GOSUB 3320 'Placement des reperes
```

```
2900 RETURN
2910 REM
2920 REM -----
2930 REM Deplacement du laser vers la gauche
2940 REM -----
2950 REM
2960 y=y-1:IF y<=0 THEN y=0:ff=1:RETURN
2970 IF x<>1 THEN IF t(x-1,y-1)=1 THEN de=2 'Vers le bas
2980 IF x<>4 THEN IF t(x+1,y-1)=1 THEN de=1 'Vers le haut
2990 IF t(x,y-1)=1 THEN bo=1 'Absorbtion
3000 RETURN
3010 REM
3020 REM -----
3030 REM Deplacement du laser vers le haut
3040 REM -----
3050 REM
3060 x=x-1:IF x<=0 THEN x=0:ff=1:RETURN
3070 IF y<>1 THEN IF t(x-1,y-1)=1 THEN de=3 'Vers la droite
3080 IF y<>9 THEN IF t(x-1,y+1)=1 THEN de=4 'Vers la gauche
3090 IF t(x-1,y)=1 THEN bo=1 'Absorbtion
3100 RETURN
3110 REM
3120 REM -----
3130 REM Deplacement du laser vers le bas
3140 REM -----
3150 REM
3160 x=x+1:IF x>=4 THEN x=5:ff=1:RETURN
3170 IF y<>1 THEN IF t(x+1,y-1)=1 THEN de=3 'Vers la droite
3180 IF y<>9 THEN IF t(x+1,y+1)=1 THEN de=4 'Vers la gauche
3190 IF t(x+1,y)=1 THEN bo=1 'Absorbtion
3200 RETURN
3210 REM
3220 REM -----
3230 REM Deplacement du laser vers la droite
3240 REM -----
3250 REM
3260 y=y+1:IF y>=9 THEN y=10:ff=1:RETURN
3270 IF x<>1 THEN IF t(x-1,y+1)=1 THEN de=2 'Vers le bas
3280 IF x<>4 THEN IF t(x+1,y+1)=1 THEN de=1 'Vers le haut
3290 IF t(x,y+1)=1 THEN bo=1 'Absorbtion
3300 RETURN
3310 REM
3320 REM -----
3330 REM Placement des reperes sur la grille
3340 REM -----
3350 REM
3360 in=x+2:x=y+1:y=in
```

```
3370 a=xp*2+7:b=yp*2+3:c=x*2+7:d=y*2+3
3380 IF yp=7 THEN b=b+1 ELSE b=b-1
3390 IF y=7 THEN d=d+1 ELSE d=d-1
3400 IF xp=11 THEN a=a+1:b=b+1
3410 IF x=11 THEN c=c+1:d=d+1
3420 IF xp=1 THEN b=b+1:a=a-1
3430 IF x=1 THEN d=d+1:c=c-1
3440 REM
3450 REM -----
3460 REM Affichage de l'entree
3470 REM -----
3480 REM
3490 le=le+1
3500 IF le=27 THEN fin=1:nf=1:RETURN
3510 LOCATE a,b
3520 PRINT CHR$(64+le)
3530 REM
3540 REM -----
3550 REM Affichage eventuel de la sortie
3560 REM -----
3570 REM
3580 IF bo=1 THEN ff=0:c=sc:RETURN
3590 LOCATE c,d
3600 PRINT CHR$(1e+96)
3610 ff=0
3620 c=sc 'Restitution de C
3630 RETURN
3640 REM
3650 REM -----
3660 REM Placement d'un trou noir
3670 REM -----
3680 REM
3690 tn=tn+1 'Un trou noir de plus
3700 LOCATE 1,4
3710 INPUT "Ligne ";li
3720 LOCATE 15,4
3730 INPUT "Colonne ";co
3740 LOCATE co*2+9,li*2+7
3750 PRINT"#
3760 LOCATE 1,4
3770 PRINT SPACE$(40)
3780 r(li,co)=1 'Memorisation
3790 GOSUB 3990 'Test de fin de partie
3800 RETURN
3810 REM
3820 REM -----
3830 REM Effacement d'un trou noir
3840 REM -----
3850 REM
```

```
3860 er=er+1 'Memorisation de l'erreur
3870 LOCATE 1,4
3880 INPUT "Ligne ";li
3890 LOCATE 15,4
3900 INPUT "Colonne ";co
3910 LOCATE co*2+9,li*2+7
3920 PRINT " "
3930 LOCATE 1,4
3940 PRINT SPACE$(40)
3950 r(li,co)=0 'Memorisation
3960 GOSUB 3990 'Test de fin de partie
3970 RETURN
3980 REM
3990 REM -----
4000 REM Test de fin de partie
4010 REM -----
4020 REM
4030 w=0
4040 FOR i=1 TO 4
4050   FOR j=1 TO 9
4060     IF t(i,j)=r(i,j) THEN w=w+1
4070   NEXT j
4080 NEXT i
4090 IF w=36 THEN fin=1:er=tn-3
4100 RETURN
4110 REM
4120 REM -----
4130 REM Commentaire de fin de partie
4140 REM -----
4150 REM
4160 CLS

4170 IF er=0 AND nf=0 THEN PRINT"Parcours sans faute. Bravo
!"
4180 IF er<>0 THEN PRINT er;"erreur(s) de parcours."
4190 IF nt=1 THEN PRINT"La partie n'est pas finie. Pas de ch
ance."
4200 RETURN
```

Le listing de la version Turbo Pascal est le suivant :

```
Program Black_Box;

Var t,
    r : array[1..4,1..9] of byte;
    a,b,c,d,i,j,xp,yp,Fin,sc,bo,de,x,y,ff,le,inn,Li,Co,Cp,W : Byte;
    ch : Char;

Procedure Presentation;
begin
    Cp:=0;
    Le:=0;
    Fin:=0;
    For I:=1 to 4 do
        For J:=1 to 9 do
            begin
                T[I,J]:=0;
                R[I,J]:=0;
            end;

    Randomize;
    For I:=1 to 3 do
        begin
            Repeat
                a:=Round(Random(2))+2;
                b:=Round(Random(2))+2;
            until (T[a,b] <> 1);
            T[a,b]:=1;
        end;
        xp:=2;
        yp:=7;
    end;

Procedure Droite;
begin
    If (xp<=9) and (yp<>1) and (yp<>4) and (yp<>5) then
        begin
            GotoXY(XP*2+7,yp*2+3);
            Write(' ');
            If (yp=3) then
                begin
                    yp:=yp-1;
                    c:=245;
                end;
            if (yp=6) then
```

```
begin
  yp:=yp+1;
  c:=244;
end;
xp:=xp+1;
GotoXY(XP*2+7,YP*2+3);
  Write(Chr(c));
end;
end;
```

Procedure Gauche;

```
begin
  If (xp>=3) and (yp<>1) and (yp<>4) and (yp<>5) then
  begin
    GotoXY(XP*2+7,yp*2+3);
    Write(' ');
    If (yp=3) then
    begin
      yp:=yp-1;
      c:=245;
    end;
    if (yp=6) then
    begin
      yp:=yp+1;
      c:=244;
    end;
    xp:=xp-1;
    GotoXY(XP*2+7,YP*2+3);
    Write(Chr(c));
  end;
end;
```

Procedure Bas;

```
begin
  If (yp>5) or ((xp<>1) and (xp<>2) and (xp<>10) and (xp<>11)) then
  else
  begin
    GotoXY(XP*2+7,yp*2+3);
    Write(' ');
    If (xp=10) then
    begin
      xp:=xp+1;
      c:=247;
    end;
  end;
```

```

    if (xp=2) then
    begin
        xp:=xp-1;
        c:=246;
    end;
    yp:=yp+1;
    GotoXY(XP*2+7,YP*2+3);
    Write(Chr(c));
end;
end;
Procedure Haut;
begin
    If (yp<4) or ((xp<>1) and (xp<>2) and (xp<>10) and (xp<>11)) then
    else
    begin
        GotoXY(XP*2+7,yp*2+3);
        Write(' ');
        If (xp=10) then
        begin
            xp:=xp+1;
            c:=247;
        end;
        if (xp=2) then
        begin
            xp:=xp-1;
            c:=246;
        end;
        yp:=yp-1;
        GotoXY(XP*2+7,YP*2+3);
        Write(Chr(c));
    end;
end;

Procedure LH;
begin
    x:=x-1;
    If (x>0) then
    begin
        if (y<>1) and (T[x-1,y-1]=1) then de:=3;
        if (y<>9) and (T[x-1,y+1]=1) then de:=4;
        if (T[x-1,y]=1) then bo:=1;
    end
    else
    begin
        x:=0;
        ff:=1;
    end;
end;

```

```
end;  
end;
```

```
Procedure LB;  
begin  
  x:=x+1;  
  If (x<4) then  
    begin  
      if (y<>1) and (T[x+1,y-1]=1) then de:=3;  
      if (y<>9) and (T[x+1,y+1]=1) then de:=4;  
      if (T[x+1,y]=1) then bo:=1;  
    end  
  else  
    begin  
      x:=5;  
      ff:=1;  
    end;  
end;  
end;
```

```
Procedure LD;  
begin  
  y:=y+1;  
  If (y<9) then  
    begin  
      if (x<>1) and (T[x-1,y+1]=1) then de:=2;  
      if (x<>9) and (T[x+1,y+1]=1) then de:=1;  
      if (T[x,y+1]=1) then bo:=1;  
    end  
  else  
    begin  
      y:=10;  
      ff:=1;  
    end;  
end;  
end;
```

```
Procedure LG;  
begin  
  y:=y-1;  
  If (y>0) then  
    begin  
      if (x<>1) and (T[x-1,y-1]=1) then de:=2;  
      if (x<>9) and (T[x+1,y-1]=1) then de:=1;  
      if (T[x,y-1]=1) then bo:=1;  
    end  
  end;  
end;
```

```

end
else
begin
  y:=0;
  ff:=1;
end;
end;

```

```

Procedure Essai_Laser;
begin
  sc:=c;
  For I:=1 to 3 do
  begin
    GotoXY(1,1);
    For J:=1 to 40 do
      Write(' ');
    end;
    GotoXY(13,1);
    Writeln('Black Box');
    Writeln(' ----- ');
    bo:=0;
    de:=c-243;
    x:=yp-2; y:=xp-1;
    Repeat
      Case de of
        1 : LH;
        2 : LB;
        3 : LD;
        4 : LG;
      end;
    until (ff=1) or (bo=1);
    inn:=x+2;
    x:=y+1;
    y:=inn;
    a:=xp*2+7;
    b:=yp*2+3;
    c:=x*2+7;
    d:=y*2+3;
    if (yp=7) then b:=b+1 else b:=b-1;
    if (y=7) then d:=d+1 else d:=d-1;
    if (xp=11) then begin
      a:=a+1;
      b:=b+1;
    end;
    if (x=11) then begin
      c:=c+1;
      d:=d+1;
    end;
  end;
end;

```

```
if (xp=1) then begin
    a:=a-1;
    b:=b+1;
end;
if (x=1) then begin
    c:=c-1;
    d:=d+1;
end;
le:=le+1;
GotoXY(a,b);
Write(Chr(64+le));
ff:=0;
If (bo<>1) then
begin
    GotoXY(c,d);
    Write(Chr(96+le));
    c:=sc;
end
else c:=sc;
end;

Procedure Demande_Tir;
begin
    Cp:=Cp+1;
    GotoXY(1,1);
    Writeln('Utilisez les touches fleches');
    Writeln('et T pour tirer. ');
    Repeat
        Repeat
            until KeyPressed;
        Read(Kbd,Ch);
        Case Ord(Ch) of
            243 : Droite;
            242 : Gauche;
            241 : Bas;
            240 : Haut;
        end;
    until (UpCase(Ch) = 'T');
    Essai_Laser;
end;

Procedure Test_Fin;
begin
    W:=0;
    For I:=1 to 4 do
```

```

    For J:=1 to 9 do
      If T[I,J]=R[I,J] then W:=W+1;
    If W=36 then Fin:=1;
end;
```

```

Procedure Placement;
begin
  GotoXY(1,4);
  Write('Ligne : ');
  Read(Li);
  Write(' Colonne : ');
  Readln(Co);
  GotoXY(co*2+9,li*2+7);
  Write('#');
  GotoXY(1,4);
  For I:=1 to 40 do Write(' ');
  R[Li,Co]:=1;
  Test_Fin;
end;
```

```

Procedure Jeu;
begin
  ClrScr;
  Writeln('          Black Box');
  Writeln('          -----');

```

```

For I:=1 to 5 do
  For J:=1 to 10 do
    begin
      GotoXY(8+2*j,6+2*i);
      Write('.');
    end;
GotoXY(1,20);
Write('T=Tir, P=Placement');
GotoXY(11,17);
Write(Chr(244));
xp:=2;
yp:=7;
c:=244;
Repeat
  Repeat
    until Keypressed;
  Read(Kbd,Ch);
  Ch:=UpCase(Ch);
  Case Ch Of

```

```
        'T' : Demande_Tir;
        'P' : Placement;
    end;
until (Fin=1);
end;
```

```
Procedure Commentaire;
begin
    ClrScr;
    Writeln('Fin du jeu en ',Cp,' coups. ');
end;
```


```
{-----}
{ Programme principal }
{-----}
```

```
begin
    Presentation;
    Jeu;
    Commentaire;
end.
```

Lancez le programme. Sur la version Basic, un écran de présentation est affiché :

Black Box

Une boîte noire renferme trois trous noirs capables de dévier ou d'absorber le rayon issu du tube à effet laser selon les deux lois suivantes :



Appuyez sur une touche . . .

Que vous utilisiez la version Basic ou Turbo Pascal l'écran de jeu est le suivant :

Black Box

```

. . . . .
. . . . .
. . . . .
. . . . .
. . . . .
^

```

T = Tir, P = Placement, E = Effacement

Déterminez la position des trous noirs en sélectionnant l'option « Tir » du menu (appuyez sur la touche T).

Le programme vous invite à choisir l'origine du tir en utilisant toutes les flèches et à activer le rayon laser en appuyant sur la touche T. Une lettre de l'alphabet est affichée en majuscule. Elle marque l'origine du tir. Si le rayon n'est pas absorbé, la même lettre apparaît en minuscule sur l'une des sorties de la boîte.

Exemple d'absorption :

Black Box

```

. . . . .
. . . . .
. . . . .
. . . . .
. . . . .

```

^
A

T = Tir, P = Placement, E = Effacement

Exemple de déviation :

Black Box

```

. . . . .
. . . . .
. . . . .
a . . . . .
. . . . .

```

^
A

T = Tir, P = Placement, E = Effacement


Utilisez l'option P du menu pour placer un trou noir, et l'option E pour en effacer un.

Lorsque les trois trous noirs ont été découverts, le programme affiche le nombre d'erreurs commises.

LE PROGRAMME EN DÉTAIL

Les versions Basic et Turbo Pascal du programme étant relativement proches, nous n'analyserons que la version Basic.

La logique du programme apparaît dans l'ordinogramme suivant :



Le programme débute par la section de programme principal qui active les trois sous-programmes principaux :

```
1040 GOSUB 1100 'Présentation
1050 GOSUB 1600 'Jeu
1060 GOSUB 4120 'Commentaire de fin de partie
```

Le sous-programme de présentation débute par la définition des tableaux de jeu :

```
1130 DIM t(4,9),r(4,9)
```

Il se poursuit par le tirage aléatoire de la position des trois trous noirs :

```
1190 FOR i=1 TO 3
1200     a=INT (RND(1)*2)+2
1210     b=INT (RND(1)*2)+2
1220     If t(a,b)=1 THEN i=i-1 ELSE t(a,b)=1
1230 NEXT i
```

Le sous-programme de présentation se termine par l'affichage de la règle du jeu sur l'écran :

```
1350 MODE 1
1360 PRINT "          Black Box"
...
1550 PRINT"Appuyez sur une touche..."
```

Le sous-programme de jeu débute par l'affichage écran de la grille :

```
1670 FOR i=1 to 5
1680     FOR j=1 to 10
1690         LOCATE 8+2*j,6+2*i
1700         PRINT "."
1710     NEXT j
1720 NEXT i
```

Il se poursuit après l'attente d'une action au clavier :

```
1840 ch$=INKEY$
1850 IF ch$=" " THEN 1840 'Attente de l'appui sur une touche
1860 ch$=UPPER$ (ch$)
```

Le programme de traitement activé dépend de la touche actionnée :

```
1870 IF ch$="T" THEN GOSUB 1930
1880 IF ch$="P" THEN GOSUB 3650
1890 IF ch$="E" THEN GOSUB 3820
```

Le sous-programme de jeu prend fin lorsque les trois trous noirs ont été trouvés :

```
1900 IF fin <> 1 THEN 1840
```

L'appui sur la touche T provoque l'activation du sous-programme de tir. Le programme se met en attente de l'appui sur une touche du clavier :

```
2010 ch$=INKEY$
2020 IF ch$=" " THEN 2010
```

Le sous-programme de traitement activé dépend de la touche pressée :

```
2030 IF ASC(ch$)=243 THEN GOSUB 2110      'Vers la droite
2040 IF ASC(ch$)=242 THEN GOSUB 2550      'Vers la gauche
2050 IF ASC(ch$)=241 THEN GOSUB 2390      'Vers le bas
2060 IF ASC(ch$)=240 THEN GOSUB 2530      'Vers le haut
2070 IF UPPER$(ch$)<>"T" THEN 2010
2080 GOSUB 2670 'Essai laser
```

Le sous-programme de déplacement vers la droite teste la position du curseur. Le déplacement n'est effectué que sous certaines conditions :

```
2150 IF xp>9 OR yp=1 OR yp=4 OR yp=5 THEN RETURN
```

Dans le cas où le déplacement est possible, le curseur est effacé :

```
2160 LOCATE xp*2+7,yp*2+3
2170 PRINT " "
```

Le caractère représentant le curseur est calculé dans la variable c :

```
2180 IF yp=3 THEN yp=yp-1:c=245
2190 IF yp=6 THEN yp=yp+1:c=244
```

et le curseur est affiché à sa nouvelle position :

```
2200 xp=xp+1
2210 LOCATE xp*2+7,yp*2+3
2220 PRINT CHR$(c)
```

Les sous-programmes de déplacement vers la gauche, le haut et le bas sont similaires. Nous n'y reviendrons pas.

Lorsque la touche T est pressée dans le sous-programme de tir, les messages affichés en haut de l'écran sont effacés :

```
2740 LOCATE 1,1
2750 PRINT SPACE$(120)
```

et l'éventuelle sortie du laser est calculée.

Le sens du déplacement initial du laser est connu grâce à la valeur de la variable c :

```
2850 de=c-243      'Sens du déplacement
```

En fonction de la valeur du déplacement, un sous-programme de calcul approprié est activé :

2870 ON de GOSUB 3020,3120,3220,2970

Les quatre sous-programmes de calcul fonctionnent selon le même principe.

Par exemple pour le programme concernant un déplacement vers le haut. Les nouvelles coordonnées du laser sont calculées. La variable ff est mise à 1 si le laser sort de la boîte :

3060 x=x-1:IF x<=0 THEN x=0:ff=1:RETURN

Lorsqu'une case voisine contient un trou noir, le rayon laser est dévié :

**3070 IF y<>1 THEN IF t(x-1,y-1)=1 THEN de=3 'Vers la droite
3080 IF y<>9 THEN IF t(x-1,y+1)=1 THEN de=4 'Vers la gauche**

Lorsque la case courante contient un trou noir, le rayon laser est absorbé :

3090 IF t(x-1,y)=1 THEN bo=1 'Absorption

Dans la boucle de calcul, le déplacement du laser se poursuit jusqu'à sa sortie ou son absorption :

2880 IF ff<>1 AND bo<>1 THEN 2870

Le sous-programme de placement des repères est ensuite activé :

2890 GOSUB 3320 'Placement des repères

La position des repères est calculée dans les variables :

- a : abscisse entrée,
- b : ordonnée entrée,
- c : abscisse sortie,
- d : ordonnée sortie :

3360 in=x+2:x=y+1:y=in

3370 a=xp*2+7:b=yp*2+3:c=x*2+7:d=y*2+3

Ces coordonnées sont ajustées en fonction de la position d'affichage (en haut, en bas, à droite ou à gauche) :

3380 IF yp=7 THEN b=b+1 ELSE b=b-1

3390 IF y=7 THEN d=d+1 ELSE d=d-1

3400 IF xp=11 THEN a=a+1:b=b+1

3410 IF X=11 THEN c=c+1:d=d+1

3420 IF xp=1 THEN b=b+1:a=a-1

3430 IF x=1 THEN d=d+1:c=c-1

L'affichage de l'entrée se fait uniquement si toutes les lettres de l'alphabet n'ont pas été utilisées :

```
3490 le=le+1
3500 IF le=27 THEN fin=1:nf=1:RETURN
3510 LOCATE a,b
3520 PRINT CHR$(64+le)
```

L'affichage de la sortie se fait uniquement si le rayon laser n'a pas été absorbé :

```
3580 IF bo=1 THEN ff=0:c=sc:RETURN
3590 LOCATE c,d
3600 PRINT chr$(le+96)
```

Les sous-programmes de placement et d'effacement de trous noirs sont relativement similaires. Par exemple pour le placement d'un trou noir, sa position est entrée :

```
3700 LOCATE 1,4
3710 INPUT "Ligne ";li
3720 LOCATE 15,4
3730 INPUT "Colonne ";co
```

et le trou noir est affiché dans la boîte :

```
3740 LOCATE co*2+9,li*2+7
3750 PRINT "#"
```

Les messages affichés en haut de l'écran sont effacés :

```
3760 LOCATE 1,4
3770 PRINT SPACE$(40)
```

La position du trou noir est mémorisée dans le tableau r :

```
3780 r(li,co)=1           'Mémorisation
```

et le sous-programme situé en ligne 3990 est activé pour déterminer si tous les trous noirs ont été découverts :

```
3790 GOSUB 3990   'Test de fin de partie
```

Le sous-programme de test de fin de partie est très simple. Il compare le contenu des tableaux t et r. Si ces deux tableaux sont identiques, la variable fin est initialisée à 1 :

```
4030 w=0
4040 FOR i=1 TO 4
4050     FOR j=1 TO 9
4060         IF t(i,j)=r(i,j) THEN w=w+1
4070     NEXT j
4080 NEXT i
4090 IF w=36 THEN fin=1:er=tn-3
```

Le sous-programme activé en fin de partie est très simple. Il commente la partie en félicitant le joueur, ou en affichant le nombre d'erreurs commises :

```
4170 IF er=0 AND nf=0 THEN PRINT "Parcours sans faute:  
                                Bravo!"  
4180 IF er<>0 THEN PRINT er;"erreur (s) de parcours."  
4190 IF nf=1 THEN PRINT "La partie n'est pas finie.  
                                Pas de chance."
```